ELLIPTICAL ORBITS: NUMERICAL SIMULATION

Link to: physicspages home page.

To leave a comment or report an error, please use the auxiliary blog.

Reference: Carroll, Bradley W. & Ostlie, Dale A. (2007), *An Introduction to Modern Astrophysics*, 2nd Edition; Pearson Education - Chapter 2, Problem 2.16.

To get an idea of what elliptical orbits look like with a few different eccentricities, we can modify our Maple version of the *Orbit* program to generate a single polar plot of one orbit, then combine the graphs onto a single plot. The algorithm for calculating the orbit is the same as before, so here's the Maple code for generating the graphs:

```
[code language=''java'']
with(plots):
G := 0.6673e-10;
AU := 0.14959787066e12;
M__sun := 0.19891e31;
rad2deg := 180/Pi;
secsYear := 365.25*(3600*24);
orbit := proc (M__strsun, a__AU, e, n, col)
local M__star, a, P, dt, t, theta, LoM, r, i, dtheta;
M__star := M__strsun*M__sun;
a := a__AU*AU;
P := sqrt(4*Pi^2*a^3/(G*M__star));
dt := P/(n-1);
t := Array(0 .. n, proc (i) options operator, arrow; i*dt end proc);
r := Array(0 .. n, datatype = float);
theta := Array(0 .. n, datatype = float);
theta[0] := 0.;
LoM := sqrt(G*M__star*a*(1-e^2));
for i from 0 to n-1 do
  r[i] := a*(1-e^2)/(1+e*cos(theta[i]));
  dtheta := LoM*dt/r[i]^2;
  theta[i+1] := evalf(theta[i]+dtheta)
end do;
r[n] := a*(1-e^2)/(1+e*cos(theta[n]));
theta := theta*rad2deg;
r := r/AU;
```

```
    polarplot(r, theta, angularunit = degrees, colour = col)
    end proc:
    plot0 := orbit(1, 1, 0, 1000, "green")
    plot4 := orbit(1, 1, .4, 1000, "red")
    plot9 := orbit(1, 1, .9, 1000, "blue")
    display([plot0, plot4, plot9])
    [/code]
```

After defining the *orbit* procedure, we call it three times, generating ellipses with a semimajor axis of 1 AU, and eccentricities of 0 (green), 0.4 (red) and 0.9 (blue), and save each result (which is the last thing calculated within the procedure in each case, so the procedure returns the *polarplot* object). Maple's *display* command on the last line combines all three plots onto a single diagram, which is this:



The centre of the plot is a focus for all three ellipses.

PINGBACKS

Pingback: Mars's orbit: ellipse versus circle